

Robin Choudhury

Datagrunnlaget for AMEN
Teknisk dokumentasjon

Notater

Innhold

1 Innledning.....	4
2 Tidsserier	5
2.1 Aggregering.....	5
2.2 Samle data	6
3 Koeffisienter	7
4 Datastruktur	9
Referanser.....	10
Vedlegg 1. Tabell- og figuroversikt	11
Vedlegg 2. Programmer og makroer	12
inndata.inp.....	12
fame2troll.src	13
kvarts2amen.src	16
lagdata.pro.....	18

1 Innledning

AMEN-modellen er utviklet spesielt med tanke på at beregningene skal kunne gjøres på personlige datamaskiner i et Windows-miljø. I likhet med blant andre KVARTS og MODAG er Portable Troll valgt som programvare for å håndtere modellen og dens omgivelser. Modellen er en kvartalsmodell og flesteparten av tidsseriene er hentet fra KVARTS-databasene. Disse databasene er i Fame-format, og kan ikke brukes når vi ønsker å kjøre modellen på en PC uten Fame. Vi har valgt en løsning som innebærer at Fame-databasene konverteres til databaser i tekstformat, nærmere bestemt Troll-formatet FORMDATA, som kan overføres mellom Troll-versjoner som kjører på forskjellige plattformer. For eksempel kan en FORMDATA database laget av Troll på en UNIX-maskin leses av Troll som kjører på en PC i et Windows-miljø. Med utgangspunkt i kvartalsdatabasene ønsker vi å bearbeide og samle sammen tilstrekkelig med data for AMEN. Med tilstrekkelig forstås at det skal være forholdsvis lett å få tilgang på andre data enn de som til enhver tid inngår direkte i modellen uten å måtte konvertere og filoverføre. Dette notatet beskriver hvordan data fra KVARTS-databasene på en UNIX-plattform overføres til databaser for AMEN beregnet på PC i et Windows-miljø¹.

¹ Rutinene som brukes ligger på området /ssb/frisch/r1/amen/amen95/lagdata/. Programmene som brukes er kompilert for bruk på Compaq-maskiner med Alpha-prosessorer.

2 Tidsserier

AMEN har et mye mer aggregert nivå en KVARTS. For eksempel er det kun fem produksjonssektorer og to kapitalarter. Første skritt på veien til å lage den databasen vi vil overføre til Windows er å lage en database som inneholder de aggregerte seriene. Neste skritt er å samle sammen tidsserier fra denne og de andre kvartalsdatabasen til en felles database. Konverteringen fra Fame-format til Troll-format skjer under denne innsamlingen av data, ved at databasen vi skriver til defineres som en Troll-database. Til slutt filoverføres denne databasen til Windows. For lett å ha tilgang på variable som kan tenkes brukt i modellen skal vi overføre alle tidsserier fra KVARTS-databasene.

2.1 Aggregering²

For å aggregere fra KVARTS-nivå til AMEN-nivå lages først en Fame-database som heter `tempkv2amen.db`. De serier som danner aggregatene vi ønsker å lage leses fra KVARTS-databasen `kv97.db`³, og skrives til `tempkv2amen.db`. Dette oppnåes ved å kjøres Fame-filen `inndata.inp`.

Vi skal nå studere nærmere hvordan aggregeringen foregår. I Figur 1 viser vi begynnelsen

Figur 1 Utdrag av filen `inndata.inp`

```
close all
over on
open <acc r> "/ssb/ovibos/fame/kvarts/kvdata97/kv97" as kvdata
open <acc o> tempkv2amen

glue ""
load lagdata

$lagdata "VA",a,kvdata,tempkv2amen
$lagdata "VA",va,kvdata,tempkv2amen
```

av filen `inndata.inp`. I de fire første linjene lukkes eventuelt åpne databaser, vi setter på overskrivingsmodus, og åpner to databaser. Databasen `/ssb/ovibos/fame/kvarts/kvdata97/kv97` er databasen vi ønsker å lese fra, mens `tempkv2amen` er den databasen vi ønsker å lagre de aggregerte seriene i. Den første databasen er åpnet for lesing, med opsijonen `<acc r>` (access read), mens den andre er åpnet for overskrivning med opsijonen `<acc o>` (access overwrite). Vi har forenklet navnet på databasen `/ssb/ovibos/fame/kvarts/kvdata97/kv97` ved å gi den et kallenavn, eller et alias, med opsijonen `as kvdata`. Når vi siden refererer til databasen kan vi bruke navnet **kvdata** fremfor hele adressen. Det er verd å merk seg at filetternavnet `.db` er utelatt, og at vi ikke oppgir hele adressen på databasen `tempkv2amen` fordi vi står på det området der vi ønsker å lagre denne.

Opsijonen `glue ""` betyr at vi ikke ønsker å sette inn tekst når vi konstruerer navnet på tidsseriene. Instruksjonen `load lagdata` gir oss tilgang til å bruke en Fame-prosedyre med navn **lagdata**. De to siste linjene i figuren viser hvordan seriene aggregeres. Kommandoen

`$lagdata "VA",a,kvdata,tempkv2amen`

² Fame inputfilen `inndata.inp`, og prosedyren `lagdata.pro`, som er sentrale her, er skrevet av Jørgen Ouren, Statistisk sentralbyrå.

³ I skrivende stund er 1997 basisår i databasene. Navnet på denne databasen vil variere med basisåret.

betyr at vi sender de fire parameterne *"VA"*, *a*, *kvdata*, *tempkv2amen* til prosedyren **lagdata**. Første parameter *"VA"* sier noe om hvilken "liste" det skal aggregeres over. Eksempler på variable som skal aggregeres over denne listen er eksport og import. En annen liste heter *"CP"* og brukes for privat konsum. Listene er definert i prosedyren *lagdata.pro*. Neste argument, *a*, er navnet på tidsserien vi ønsker å lage. De to siste parameterne er henholdsvis aliaset for databasen vi leser fra (**kvdata**) og navnet på databasen vi skriver til (*tempkv2amen*).

For å utføre aggregeringen gjør vi følgende:

1. Logg deg på en Compaq-maskin og gå til området */ssb/frisch/r1/amen/amen95/lagdata/*
2. Start Fame ved å skrive *ile fame* (evt. *ffame*)
3. Skriv *input inndata*
4. Avslutt Fame ved å skrive *exit*

Nå er det opprettet en database som heter *tempkv2amen.db*. Merk at databasen overskrives hver gang *inndata.inp* brukes.

2.2 Samle data

Som tidligere nevnt ønsker vi å overføre alle tidsserier fra kvartalsdatabasene i tillegg til de aggregerte tidsseriene vi har laget. Alle tidsseriene blir samlet sammen fra sine respektive Fame-databaser til en database som blir opprettet. Denne filoverføres senere til Windows. Programmet som brukes til å samle sammen data er skrevet i Troll og heter *kvarts2amen.prg*. Det utfører følgende:

1. Åpner kvartalsdatabasene *kv97.db*, *inr97.db*, *dummy.db* og *tempkv2amen.db* for lesing
2. Oppretter databasen *kvarts2amen.frm* for skriving
3. Leser alle tidsserier fra databasene under punkt 1, skriver til databasen under punkt 2.

For basisår 1997 er adressene følgende:

```
/ssb/ovibos/fame/kvarts/kvdata97/kv97.db
/ssb/ovibos/fame/kvarts/inr/inr97.db
/ssb/ovibos/fame/kvarts/inr/dummy.db
/ssb/frisch/r1/amen/amen95/lagdata/tempkv2amen.db
```

For å samle sammen data gjør vi følgende:

1. Logg deg på en Compaq-maskin og gå til området */ssb/frisch/r1/amen/amen95/lagdata/*
2. Start Troll ved å skrive *ile troll*
3. Skriv *&kvarts2amen*
4. Avslutt Troll med å skrive *trexit*

For hver av databasene som er ferdig overført til den databasen vi samler alle data i skriver programmet ut en melding om dette, samt hvor mange serier som er overført. Databasen *kvarts2amen.frm* er nå ferdig og kan filoverføres til Windows.

3 Koeffisienter

Som nevnt i "Brukerveiledning for AMEN", se Choudhury (2001), brukes hele koeffisientdatabasen fra KVARTS i AMEN. Dette er en Fame-database som må konverteres til en portabel database. Vi bruker et Troll-program som heter `fame2troll.prg` til dette. Fremgangsmåten er følgende:

1. Kopier koeffisientdatabasen til området `/ssb/frisch/r1/amen/amen95/lagdata/`
2. Start Troll ved å skrive `ile troll`
3. Skriv `&fame2troll`
4. Velg *n* (for ny Troll-database)
5. Skriv navnet på koeffisientfilen (den som ble kopiert til dette området og er i Fame-format)
6. Skriv inn navnet på den nye koeffisientfilen (den som skal bli en FORMDATA-fil)
7. Skriv *q* for å avslutte programmet
8. Skriv `trex` for å avslutte Troll

Et eksempel på bruk av programmet er vist i Figur 2. Vi har først kopiert koeffisientdatabasen `kv97con.db` fra KVARTS-området til dette området (se punkt 1 over). Vi starter Troll fra en UNIX kommandolinje ved å skrive `ile troll`. Så starter vi programmet ved å skrive `&fame2troll` og vi får opp

Figur 2 Konvertering av koeffisientdatabasen fra Fame format til FORMDATA format

```
TROLL Release 1.09
Copyright (C) Intex Solutions, Inc. 1993-2000
Copyright (C) Massachusetts Institute of Technology 1978-1987

TROLL Command: &fame2troll

    Dette programmet overfører data fra en FAME-database til en
    FORMDATA-database. Gyldige kommandoer er:

    Q: Avslutter programmet
    N: Lag ny TROLL-database
    F: List opp FAME-databaser på dette arkivet
    T: List opp FORMDATA-databaser på dette arkivet
    >n

Skriv navn på FAME-database (eller 'Q' for å avslutte): kv97con

Skriv navn på TROLL-database: (eller 'Q' for å avslutte): kv97contest2

Ferdig med å overføre data fra kv97con.db til kv97contest2.frm.

    Dette programmet overfører data fra en FAME-database til en
    FORMDATA-database. Gyldige kommandoer er:

    Q: Avslutter programmet
    N: Lag ny TROLL-database
    F: List opp FAME-databaser på dette arkivet
    T: List opp FORMDATA-databaser på dette arkivet
    >q

Programmet ble avbrutt av brukeren.

TROLL Command: trexit
```


en meny der vi velger *n* for å lage en ny Troll-database. Deretter skriver vi navnet på databasen vi ønsker å konvertere, i dette tilfellet `kv97con`. Så skriver vi navnet på Troll-databasen vi ønsker å legge de konverterte koeffisientene i. Her har vi valgt å kalle den `kv97contest2`. Det er ikke nødvendig å

skrive filetternavnet .db og .frm, da programmet forutsetter at det er en Fame-database (.db) som skal konverteres til en Troll-database (.frm). Det er mulig å bruke samme filnavn på den databasen vi ønsker å lagre de konverterte koeffisientene i som den vi konverterer fra. Databasene vil uansett få forskjellige filetternavn. Når programmet er ferdig med å overføre data skrives det en melding om dette på skjermen, og menyen kommer frem på nytt. Vi skriver *q* for å avslutte dette programmet, og *texit* for å avslutte Troll.

4 Datastruktur

I Figur 3 viser vi strukturen på programmer/makroer og databaser som skal aggregeres, samles, og til slutt overføres til Windows. Tidsseriene samles i databasen *kvarts2amen.db* som filoverføres. *kv97.db* brukes i tillegg til å lage noen aggregerte tidsserier i databasen *tempkv2amen.db*. Koeffisientdatabasen *kv97con.db* konverteres f.eks. til *kv97contest2.frm* og filoverføres.

Figur 3 Datastruktur (UNIX)



Tabell 1 Forklaring av symboler

	Boks	Pil
Heltrukken	database	skrive til
Stiplet	program/makro	lese fra

Filnavn skrevet i kursiv er valgfrie.

Referanser

Choudhury, R. (2001): *Brukerveiledning for AMEN*. Kommer i serien Notater, Statistisk sentralbyrå.

Hollinger, P. and L. Spivakovsky (1993): Portable TROLL 0.86. Reference Manual, Intex Solutions, Inc.

Hollinger, P. and L. Spivakovsky (1996): Portable TROLL Programming Language for Release 1.0, Intex Solutions, Inc.

FAME Software Corporation (1990); User's guide to FAME.

Tabell- og figuroversikt

Tabelloversikt

<i>Tabell 1 Forklaring av symboler.....</i>	<i>9</i>
---	----------

Figuroversikt

<i>Figur 1 Utdrag av filen inndata.inp.....</i>	<i>5</i>
<i>Figur 2 Konvertering av koeffisientdatabasen fra Fame format til FORMDATA format.....</i>	<i>7</i>
<i>Figur 3 Datastruktur (UNIX).....</i>	<i>9</i>

Programmer og makroer

Her følger en utskrift av programmer og makroer som brukes.

inndata.inp

```
close all
over on
open <acc r> "/ssb/ovibos/fame/kvarts/kvdata97/kv97" as kvdata
open <acc o> tempkv2amen

glue ""
load lagdata

$lagdata "VA",a,kvdata,tempkv2amen
$lagdata "VA",va,kvdata,tempkv2amen
$lagdata "VA",ab,kvdata,tempkv2amen
$lagdata "VA",vab,kvdata,tempkv2amen
$lagdata "VA",i,kvdata,tempkv2amen
$lagdata "VA",vi,kvdata,tempkv2amen
$lagdata "VA",xvb,kvdata,tempkv2amen
$lagdata "VA",vxvb,kvdata,tempkv2amen
$lagdata "VA",toll,kvdata,tempkv2amen
$lagdata "VA",vtoll,kvdata,tempkv2amen

$lagdata "CP",c,kvdata,tempkv2amen
$lagdata "CP",vc,kvdata,tempkv2amen

$lagdata "PS",x,kvdata,tempkv2amen
$lagdata "PS",vx,kvdata,tempkv2amen
$lagdata "PS",q,kvdata,tempkv2amen
$lagdata "PS",y,kvdata,tempkv2amen
$lagdata "PS",h,kvdata,tempkv2amen
$lagdata "PS",vh,kvdata,tempkv2amen

$lagdata "PS",fd,kvdata,tempkv2amen

$lagdata "PS",yd,kvdata,tempkv2amen
$lagdata "PS",ye,kvdata,tempkv2amen
$lagdata "PS",yf,kvdata,tempkv2amen
$lagdata "PS",yt,kvdata,tempkv2amen
$lagdata "PS",yts,kvdata,tempkv2amen
$lagdata "PS",yw,kvdata,tempkv2amen
$lagdata "PS",ywt,kvdata,tempkv2amen
$lagdata "PS",yww,kvdata,tempkv2amen
$lagdata "PS",nw,kvdata,tempkv2amen
$lagdata "PS",lw,kvdata,tempkv2amen

$lagdata "JS",JK,kvdata,tempkv2amen
$lagdata "JS",VJK,kvdata,tempkv2amen

$lagdata "KAP",fd,kvdata,tempkv2amen

$lagdata "KAP",K,kvdata,tempkv2amen

--$lagdata "PS",z,kvdata,tempkv2amen
```

fame2troll.src

```

/*****
Prosjekt .....: Amen
Program navn .....: fame2troll
Skrevet av .....: Robin Choudhury
Dato .....: 01.02.99
Versjon .....: 1.0
Programmets funksjon ..: Overføre en FAME-database til en TROLL FORMDATA-database
Filer inn .....: FAME - database (*.db)
Filer ut .....: troll - database (*.frm)
Endret når .....: DD.MM.ÅÅ
Endret av .....:
Grunn til endring ....:
*****/

addfun main;
procedure main()
{

    >>delaccess all;
    try_again:

ssst=  ("\\n\\tDette programmet overfører data fra en FAME-database til en
\\tFORMDATA-database. Gyldige kommandoer er:\\n\\t
\\tQ: Avslutter programmet\\t
\\tN: Lag ny TROLL-database\\t
\\tF: List opp FAME-databaser på dette arkivet\\t
\\tT: List opp FORMDATA-databaser på dette arkivet\\t
\\t>");

get svar ssst ;

if (svar == "Q" or svar == "q") then
{
    clear();
    print ("\\nProgrammet ble avbrutt av brukeren.\\n");
    exit();
}

if (svar == "N" or svar == "n") then
{
    fdb_navn:

    get famedata "\\n Skriv navn på FAME-database (eller 'Q' for å avslutte): ";

    if (famedata == "Q" or famedata == "q") then
    {
        clear();
        print ("\\nProgrammet ble avbrutt av brukeren.\\n");
        exit();
    }

    extent=substr(famedata,-3);

    if (extent==".DB" or extent==".db") then
    {
        fametemp=lower(famedata);
    }
    else
    {
        fametemp=lower(famedata||".db");
    }

    if (hfexist(fametemp) <> TRUE) then
    {
        print ("\\n FEIL: Databasen '",fametemp, "' finnes ikke. Prøv igjen. ");
        goto fdb_navn;
    }

    tdb_navn:

    get trolldata "\\n Skriv navn på TROLL-database: (eller 'Q' for å avslutte): ";

```

```

if (trollldata == "Q" or trollldata == "q") then
{
  clear();
  print ("\nProgrammet ble avbrutt av brukeren.\n");
  exit();
}

extent=substr(trollldata,-4);

if (extent==".FRM" or extent==".frm") then
{
  trollltemp=lower(trollldata);
}
else
{
  trollltemp=lower(trollldata||".frm");
}

if (hfexist(trollltemp) == TRUE) then
{
  get over "\n ADVARSEL: Databasen eksisterer allerede. Velg \n\tO: Overskriv\n\tN:
velg nytt navn\n\t >";

  if (over <> "O" and over <> "o" and over <> "N" and over <> "n") then
  {
    clear();
    print ("\nFEIL: Ugyldig valg. Programmet ble avbrutt.\n");
    exit();
  }

  if (over == "O" or over == "o") then
  {
    >>access fdata type fame id &(fametemp) mode r;
    >>access tdata type formdata id &(trollltemp) mode w;
    >>search data fdata;
    >>search data tdata w;

    fname=dflist("fdata",>);
    nrfname=nvals(fname);

    for (i=1; i<=nrfname; i=i+1)
    {
      nrf=fname[i];
      >>dofile &(nrf)=&(nrf);
    }

    >>delaccess all;

print("\nFerdig med å overføre data fra '",fametemp, "' til '",trollltemp,"'.\n");
  }

  if (over == "N" or over == "n") then
  {
    goto tdb_navn;
  }
}

else
{
  >>access fdata type fame id &(fametemp) mode r;
  >>access tdata type formdata id &(trollltemp) mode w;
  >>search data fdata;
  >>search data tdata w;

  fname=dflist("fdata",>);

  nrfname=nvals(fname);

  for (i=1; i<=nrfname; i=i+1)
  {
    nrf=fname[i];
    >>dofile &(nrf)=&(nrf);
  }

print("\nFerdig med å overføre data fra ",fametemp, " til ",trollltemp,".\n");

```

```

    }

    goto try_again;
}

if (svar == "F" or svar == "f") then
{
    print("_____ ");
    print("Liste over FAME-databaser på dette arkivet: ");
    print("");
    host("ls -l -r -t *.db");
    print("_____ ");
    goto try_again;
}

if (svar == "T" or svar == "t") then
{
    print("_____ ");
    print("Liste over FORMDATA-databaser på dette arkivet: ");
    print("");
    host("ls -l -r -t *.frm");
    print("_____ ");
    goto try_again;
}

else if (svar <> "N" and svar <> "n" and svar <> "Q" and svar <> "q" and svar <> "F" and
svar <> "f" and svar <> "T" and svar <> "t") then
{
    print ("\nFeil valg: '", svar, "' prøv igjen.");
    goto try_again;
}

>>delsave all;
}

```

kvarts2amen.src

```

/*****
Project .....: AMEN
Program name .....: kvarts2amen.prg
Written by .....: Robin Choudhury
Date .....: 2.8.00
Version .....: 1.0
Program function .....: Collect data series from the accessed databases
Files in .....: Fame - databases (*.db)
Files out .....: Troll - database (*.frm)
Changed when .....: DD.MM.YY
Changed by whome .....:
Reason for change .....:

```

Description.....:

```

*****/
```

```

addfun main;
procedure main()
{

    >>delaccess all;

    >>access kvdata type fame id /ssb/ovibos/fame/kvarts/kvdata97/kv97 mode r;
    >>access inrdata type fame id /ssb/ovibos/fame/kvarts/inr/inr97.db mode r;
    >>access dum type fame id /ssb/ovibos/fame/kvarts/inr/dummy mode r;
    >>access agg type fame id tempkv2amen.db mode r;

    >>search data kvdata;
    >>search data inrdata;
    >>search data dum;
    >>search data agg;

    >>access amendata type formdata id kvarts2amen.frm mode w;
    >>search data amendata w;

    kvname=dflist("kvdata",>");
    nrkvname=nvals(kvname);

    print("\t\nOverfører data. Vennligst vent ...");

    for (i=1; i<=nrkvname; i=i+1)
    {
        nrf=kvname[i];
        >>dofile &nrf=&nrf;
    }

    >>delaccess kvdata;

    print("\tFerdig med kv97. " , nrkvname, " dataserier er overført.");

    inrname=dflist("inrdata",>");
    nrinrname=nvals(inrname);

    for (i=1; i<=nrinrname; i=i+1)
    {
        nrf=inrname[i];
        >>dofile &nrf=&nrf;
    }

    >>delaccess inrdata;

    print("\tFerdig med inr97. " , nrinrname, " dataserier er overført.");

    dumname=dflist("dum",>");
    nrdumname=nvals(dumname);

    for (i=1; i<=nrdumname; i=i+1)
    {
        nrf=dumname[i];
        >>dofile &(nrf)=&(nrf);
    }
}

```

```

    }

    >>delaccess dum;

    print("\tFerdig med dummy. " , nrdummyname, " dataserier er overført.");

    fname=dflist("agg",>");
    nrfname=nvals(fname);

    for (i=1; i<=nrfname; i=i+1)
    {
        nrf=fname[i];
        >>dofile &nrf=&nrf;
    }

    print("\tFerdig med tempkv2amen. " , nrfname, " dataserier er overført.");

    >>delaccess agg;

    print("\tFerdig. Totalt " , nrkvname+nrinrname+nrdummyname+ nrfname, " dataserier er
overført.\n");

}

```

lagdata.pro

```
procedure $lagdata
argument liste,vari,fra,til
over on
glue ""

if liste eq "VA"
    til'vari&64 = fra'vari&66+fra'vari&67+fra'vari&69+fra'vari&68
    til'vari&65= fra'vari&65
    til'vari&83= fra'vari&83
    til'vari&35= fra'vari&35
    til'vari&90 = fra'vari&92s+fra'vari&93s+fra'vari&94s+fra'vari&95s &&
        +fra'vari&93k+fra'vari&94k+fra'vari&95k
    til'vari = fra'vari
    til'vari&98 = til'vari-til'vari&64-til'vari&65 &&
        -til'vari&83 -til'vari&35 -til'vari&90

else if liste eq "CP"
    til'vari&50 = fra'vari&50
    til'vari = fra'vari
    til'vari&98 = til'vari -til'vari&50

else if liste eq "JS"
    til'vari&1064 = fra'vari&1064
    til'vari&s64 = fra'vari&s64
    til'vari&0064 = til'vari&s64 - til'vari&1064

    til'vari&1065 = fra'vari&1065
    til'vari&s65 = fra'vari&s65
    til'vari&0065 = til'vari&s65-til'vari&1065

    til'vari&1083 = fra'vari&1083
    til'vari&s83 = fra'vari&s83
    til'vari&0083 = til'vari&s83-til'vari&1083

    til'vari&1090 = fra'vari&"1092s"+fra'vari&"1093k" +fra'vari&"1093s" &&
        +fra'vari&"1094k" +fra'vari&"1094s"+fra'vari&"1095k" +fra'vari&"1095s"
    til'vari&s90 = fra'vari&"s92s"+fra'vari&"s93k" +fra'vari&"s93s" &&
        +fra'vari&"s94k" +fra'vari&"s94s"+fra'vari&"s95k" +fra'vari&"s95s"
    til'vari&0090 = til'vari&s90-til'vari&1090
    til'vari&10 = lsum(wildlist(fra,string(name(vari))+"10?") except {vari&10}) -- huff!
-- jk10 og vjk10 beregnes utenfor til'vari&10 = fra'vari&10
    til'vari&1098 = til'vari&10-til'vari&1064-til'vari&1065 -til'vari&1083 &&
        -til'vari&1090
    til'vari = fra'vari
    til'vari&s98 = til'vari -til'vari&s64-til'vari&s65-til'vari&s83-til'vari&s90
    til'vari&0098 = til'vari&s98 -til'vari&1098
    til'vari&00 = til'vari -til'vari&10
else if liste eq "KAP"
    til'vari&1064 = fra'vari&1064
    til'vari&64 = fra'vari&64
    til'vari&0064 = til'vari&64 - til'vari&1064

    til'vari&1065 = fra'vari&1065
    til'vari&65 = fra'vari&65
    til'vari&0065 = til'vari&65-til'vari&1065

    til'vari&1083 = fra'vari&1083
    til'vari&83 = fra'vari&83
    til'vari&0083 = til'vari&83-til'vari&1083

    til'vari&1090 = fra'vari&"1092s"+fra'vari&"1093k" +fra'vari&"1093s" &&
        +fra'vari&"1094k" +fra'vari&"1094s"+fra'vari&"1095k" +fra'vari&"1095s"
    til'vari&90 = fra'vari&"92s"+fra'vari&"93k" +fra'vari&"93s" &&
        +fra'vari&"94k" +fra'vari&"94s"+fra'vari&"95k" +fra'vari&"95s"
    til'vari&0090 = til'vari&90-til'vari&1090
    til'vari&10 = lsum(wildlist(fra,string(name(vari))+"10?")) -- huff!
-- jk10 og vjk10 beregnes utenfor til'vari&10 = fra'vari&10
    til'vari&1098 = til'vari&10-til'vari&1064-til'vari&1065 -til'vari&1083 &&
        -til'vari&1090
    til'vari = fra'vari
    til'vari&98 = til'vari -til'vari&64-til'vari&65-til'vari&83-til'vari&90
    til'vari&0098 = til'vari&98 -til'vari&1098
    til'vari&00 = til'vari -til'vari&10
```

```

else if liste eq "PS"
  til'vari&64 = fra'vari&64
  til'vari&65= fra'vari&65
  til'vari&83= fra'vari&83
  til'vari&90 = fra'vari&"92s"+fra'vari&"93k" +fra'vari&"93s" &&
    +fra'vari&"94k" +fra'vari&"94s"+fra'vari&"95k" +fra'vari&"95s"
  til'vari = fra'vari
  til'vari&98 = til'vari-til'vari&64-til'vari&65-til'vari&83-til'vari&90
else
  type "FEIL: Ukjent liste!"
end if

end procedure

```

De sist utgitte publikasjonene i serien Notater

- | | | | |
|---------|--|---------|---|
| 2000/80 | J. Kittelsen og P. O. Lande: Forbruksundersøkinga 2000. Systemdokumentasjon. 156s. | 2001/15 | O. Klungsøyr: Markovkjede Monte Carlo i varianstkomponentmodell for sysselsettingsdata. 30s. |
| 2000/81 | J.T. Lind: Testing av stokastiske individuelle effekter i paneldatamodeller. 17s. | 2001/16 | M. Bråthen og T. Pedersen: Tilpasning på arbeidsmarkedet for personer som går ut av status som yrkeshemmet i SOFA- søkerregisteret - 1998. 27s. |
| 2001/2 | D.Q. Pham: Innføring i tidsserier - sesongjustering og X-12-AMIRA. 110s. | 2001/17 | T. Martinsen: Statistikk over energibruk i Statistisk sentralbyrå - evaluering, brukerbehov og forutsetninger. 87s. |
| 2001/3 | O. Rognstad: Eiendomsomsetning. Dokumentasjon av datagrunnlag og bearbeidingsrutine. 72s. | 2001/18 | L. Vågane: Undersøkelse om holdninger til frukt- og grøntabonement blant foreldre med barn i grunnskolen. Dokumentasjonsrapport. 26s. |
| 2001/4 | T. Nøtnæs: Innføring i kognitiv kartlegging. 20s. | 2001/19 | H. Madsen og A. Langørgen: Anslag over antall etterspørrere av grunnskoleopp-læring for voksne. 23s. |
| 2001/5 | T. Bye, M. Hansen og B. Strøm: Hvordan framskrive utslipp av klimagasser? 16s. | 2001/20 | B. Indahl, D.E. Sommervoll og J. Aasness: Virkninger på forbruksmønster, leve-standard og klimagassutslipp av endringer i konsumentpriser. 27s. |
| 2001/6 | A. Langørgen og R. Aaberge: KOM-MODE II estimert på data for 1998. 16s. | 2001/21 | A. Barstad: På vei mot det gode samfunn? Utredning til Finansdepartementet i forbindelse med arbeidet med nytt Langtidsprogram, 2002-2005. 363s. |
| 2001/7 | B.R. Joneid og J. Lajord: FD - Trygd: Dokumentasjonsrapport. Stønader til enslig forsørger. 1992-1999. 39s. | 2001/23 | L. Østby: Beskrivelse av nyankomne flykningers vei inn i det norske samfunnet. Notat til Lovutvalget som skal utrede og lage forslag til lovgivning om stønad for nyankomne innvandrere. 32s. |
| 2001/8 | T. Karlsen, E. Karstensen og E. Evensen: Beregningsrutiner og teknisk programstruktur for fylkesfordelt nasjonalregnskap. 27s. | 2001/24 | T. Nøtnæs: Innføring i bruk av fokus-grupper. 22s. |
| 2001/9 | L. Rognstad, N.M. Stølen, T. Jakobsen og P. Schønning: Regional statistikk og analyse - strategi og prioriteringer. 45s. | 2001/25 | J. Fosen, A.G. Hustoft og B.O. Lagerstrøm: Ny spørresekvens for å identifisere husholdninger i utvalgsundersøkelser. 29s. |
| 2001/10 | A. Akselsen og B.R. Joneid: FD - Trygd: Dokumentasjonsrapport. Pensjoner. Grunn- og hjelpestønader. 1992-1998. 94s. | 2001/26 | H.C. Hougen: Undersøkelse om folat-kunnskap blant kvinner i fertil alder: Dokumentasjonsrapport. 17s. |
| 2001/11 | B. Mathisen: Flyktninger og arbeidsmarkedet 4. kvartal 1999. 34s. | 2001/27 | Ø. Kleven og O.F. Vaage: Medieundersøkelsen 1999: Dokumentasjonsrapport. 49s. |
| 2001/12 | A. Rognan og N. Barrabés: NUS2000. Dokumentasjonsrapport. 36s. | | |
| 2001/13 | K.I. Bøe, J. Johansen og Ø. Sivertstøl: FD - Trygd: Dokumentasjonsrapport. Attføringspenger, 1992-1998. 88s. | | |
| 2001/14 | O. Klungsøyr: Ekstremverdimodell for industrinæringenes investeringer i 90-årene. 30s. | | |